Lundi 5 septembre 2005

Equations

Bernard Vuilleumier

Lorsqu'on recherche des solutions qui sont des nombres entiers ou rationnels dans une équation, on l'appelle équation diophantienne. Ce sujet est étudié depuis Diophante qui vécut à Alexandrie au III^{e} siècle de notre ère. L'étude de ces équations s'avère difficile et beaucoup de problèmes diophantiens ont joué un rôle important dans l'histoire des mathématiques. Ils sont, aujourd'hui encore, considérés comme de grands classiques. Le critère permettant de décider si une équation diophantienne a une infinité de solutions ou non repose par exemple sur une conjecture (de Birch et Swinnerton-Dyer) non démontrée à ce jour. Cette question figure parmi les sept problèmes du troisième millénaire dotés d'un prix d'un million de dollars !



Fig. 1 : L'équation $1 + 12x + 2x^2 + 7y + 5xy + 2y^2 = 0$ possède une infinité de solutions. Ce sont les coordonnées des points appartenant aux hyperboles partiellement représentées ci-dessus. Mais cette équation ne possède que trois solutions entières données par les coordonnées des points rouges.

Prochaine publication : lundi 3 octobre 2005

Activités proposées

1. Résoudre une équation

Résolvez l'équation ci-dessous :
a) $ax^2 + bx + c = 0$
b) Exprimez, à l'aide des coefficients a, b et c, la condition pour que l'équation possède :
• deux solutions réelles distinctes
• deux solutions réelles confondues
• deux solutions complexes

2. Résoudre un système d'équations

Résolvez le système d'équations ci-dessous :

$$x^2 + y^2 = 1$$
$$\frac{1}{2}x^2 + 2y^2 = \frac{3}{2}$$

a) pour x et y
b) en éliminant d'abord y et en résolvant ensuite pour x
c) Dessinez les courbes correspondant à ces deux équations

3. Résoudre une équation diophantienne

Résolvez les équations ci-dessous pour des valeurs entières de x et de y :
a) $1 + 12\,x + 2\,x^2 + 7\,y + 5\,x\,y + 2\,y^2 = 0$
b) $7 + 5x + x^2 + 7y + 3xy + y^2 = 0$
c) $x^2 - 61\,y^2 = 1$
d) $x^2 - 2xy + y^2 + 5x - 7y = 22$
e) $23\,x^2 + 17\,y^2 = 16933394294659350729128029263679225728 00$ avec $x > 0$ et $y > 0$
f) $x^3 - 4xy^2 + y^3 = 1$

Pour en savoir plus
• *Les dossiers de La Recherche n° 20. Août - Octobre 2005*. Mathématiques. Nouveaux défis et vieux casse-tête.
• http://www.claymath.org/millennium/
• http://perso.wanadoo.fr/yoda.guillaume/Vocabula/GlosE/EquaDiop.htm

Équations : corrigé

■ 1. Résoudre une équation

a) La commande Solve permet de résoudre des équations linéaires et polynômiales :

```
Solve[a * x ^ 2 + b * x + c == 0, x]
```

N. B. L'égalité est notée à l'aide d'un double signe égal dans *Mathematica*.

b) La valeur du déterminant $\Delta = b^2 - 4\,a\,c$ donne la condition cherchée :
• $\Delta > 0$ deux solutions réelles distinctes
• $\Delta = 0$ deux solutions réelles confondues
• $\Delta < 0$ deux solutions complexes

■ 1. Résoudre un système d'équations

a) La syntaxe suivante permet de résoudre le système par rapport à x et à y :

```
Solve[{x ^ 2 + y ^ 2 == 1, 1 / 2 x ^ 2 + 2 y ^ 2 == 3 / 2}, {x, y}]
```

b) En omettant les accolades, y est d'abord éliminé et le système est résolu par rapport à x :

```
Solve[{x ^ 2 + y ^ 2 == 1, 1 / 2 x ^ 2 + 2 y ^ 2 == 3 / 2}, x, y]
```

c) Les courbes peuvent être dessinées à l'aide de :

```
Needs["Graphics`ImplicitPlot`"]
ImplicitPlot[{x ^ 2 - y ^ 2 == 1, 1 / 2 x ^ 2 + 2 y ^ 2 == 3 / 2}, {x, -2, 2}];
```

■ 3. Résoudre une équation diophantienne

Mathematica peut résoudre les équations diophantiennes quadratiques à deux variables.
La forme générale d'une telle équation est la suivante :

$$\Phi(x, y) = a\,x^2 + b\,x\,y + c\,y^2 + d\,x + e\,y + f = 0$$

La solution dépend du déterminant $\Delta = b^2 - 4\,a\,c$ de la forme quadratique.

Équations de type hyperbolique avec déterminant carré

a) Si $\Delta > 0$ et si $\sqrt{\Delta}$ est un entier, alors l'équation possède un nombre fini de solutions et *Mathematica* fournit les solutions entières de cette équation :

```
eq = 1 + 12 x + 2 x ^ 2 + 7 y + 5 x * y + 2 y ^ 2;
(b ^ 2 - 4 a * c) /. {a -> Coefficient[eq, x ^ 2],
  b -> Coefficient[eq, x * y], c -> Coefficient[eq, y ^ 2]}
Sqrt[%]
Reduce[eq == 0, Integers]
```

```
3
```

```
(x == -4 && y == -1) || (x == 2 && y == -3) || (x == 4 && y == -9)
```

Équations de type hyperbolique avec déterminant non carré

b) Si $\Delta > 0$ et si $\sqrt{\Delta}$ n'est pas un entier, alors l'équation est une équation de type « Pell » (une équation de Pell est une équation de la forme $x^2 - D y^2 = 1$ où D n'est pas un carré). L'ensemble des solutions est vide ou infini, mais il peut être paramétré à l'aide d'un entier. Vérifions que la deuxième équation est de ce type :

```
Remove[eq]
eq = 7 + 5 x + x^2 + 7 y + 3 x * y + y^2;
b^2 - 4 a * c /. {a -> Coefficient[eq, x^2],
  b -> Coefficient[eq, x * y], c -> Coefficient[eq, y^2]}
Sqrt[%]
```

5

$\sqrt{5}$

Les solutions de ce type d'équations font apparaître le paramètre C[1] :

```
sol = Reduce[eq == 0, Integers]
```

Même si l'expression des solutions de ce type d'équations fait usage de nombres non rationnels, ces solutions sont en fait des entiers comme on peut le vérifier en attribuant une valeur entière au paramètre C[1] et en simplifiant :

```
Simplify[sol /. C[1] → 1]
```

```
(y == -129 && (x == 335 || x == 47)) || (y == 337 && (x == -885 || x == -131)) ||
 (y == 49 && (x == -131 || x == -21)) || (y == 2311 && (x == -885 || x == -6053)) ||
 (y == -6051 && (x == 2309 || x == 15839)) || (y == -883 && (x == 335 || x == 2309))
```

c) L'équation $x^2 - 61\, y^2 = 1$ est une équation de Pell. Les solutions des équations de Pell peuvent être compliquées même pour de petits coefficients :

```
sol = Reduce[x^2 - 61 y^2 == 1, Integers]
```

Attribuons une valeur entière au paramètre C[1] et simplifions pour obtenir des entiers :

```
sol /. C[1] → 2 // Simplify
```

```
(y == 798920165762330040 || y == -798920165762330040) &&
 (x == -6239765965720528801 || x == 6239765965720528801)
```

Équations de type parabolique

Si $\Delta = 0$, l'équation est de type parabolique. Si elle possède des solutions entières, il y en a une infinité. L'ensemble des solutions peut être paramétré à l'aide d'un entier C[1].

```
Remove[eq]
eq = x^2 - 2 x * y + y^2 + 5 x - 7 y;
b^2 - 4 a * c /. {a -> Coefficient[eq, x^2],
  b -> Coefficient[eq, x * y], c -> Coefficient[eq, y^2]}
sol = Reduce[eq == 22, Integers];
sol /. C[1] → 1 // Simplify
```

0

```
(x == -2 && y == -4) || (x == 4 && y == 1)
```

Équations de type elliptique

Si $\Delta < 0$, l'équation est de type elliptique. Comme une ellipse est un ensemble borné, le nombre de solutions doit être fini. Essayer de trouver toutes les solutions est impraticable pour des raisons de temps : il y a plus de 8×10^{18} valeurs entières positives pour x pour cette ellipse. Pour de grandes ellipses Reduce ne donne qu'une partie des solutions. Ici, Reduce trouve des solutions constituées de nombres entiers positifs pour x et pour y :

```
Remove[eq]
eq = 23 x^2 + 17 y^2 ;
b^2 - 4 a*c /. {a -> Coefficient[eq, x^2],
  b -> Coefficient[eq, x*y], c -> Coefficient[eq, y^2]}
sol = Reduce[eq == 1693339429465935072912802926367922572800 && x > 0 && y > 0,
  Integers]
```

```
(x == 1234567890987654321 && y == 9876543210123456789) ||
 (x == 2394388915976549628 && y == 9583927013507483052) ||
 (x == 3587688774846621081 && y == 9066079904941225629) ||
 (x == 4628858032573225308 && y == 8403549375095756172) ||
 (x == 4730542803202013073 && y == 8326586121887736693) ||
 (x == 6448688263945408950 && y == 6583719143723572530) ||
 (x == 6563464511756847993 && y == 6428433631978684413) ||
 (x == 7787179624084878150 && y == 4191136305399154530)
```

Équations de Thue

Une équation de Thue est une équation diophantienne de la forme :

$$\Phi(x, y) = m$$

où $\Phi(x, y)$ est une forme irréductible de degré ≥ 3. Le nombre de solutions d'une équation de Thue est toujours fini. *Mathematica* peut en principe résoudre n'importe quelle équation de Thue, mais le temps nécessaire pour trouver les solutions croît très vite avec le degré et la taille des coefficients. La partie la plus difficile pour l'algorithme est de calculer une borne pour la taille des solutions. Si l'entrée contient une inégalité bornant la taille des solutions, ces dernières peuvent être calculées beaucoup plus rapidement par *Mathematica*.

```
Reduce[x^3 - 4 x *y^2 + y^3 == 1, {x, y}, Integers] // Timing
```

```
{3.62184 Second, (x == -2 && y == 1) || (x == 0 && y == 1) ||
  (x == 1 && y == 0) || (x == 1 && y == 4) || (x == 2 && y == 1) || (x == 508 && y == 273)}
```

```
Reduce[x^3 - 4 x *y^2 + y^3 == 1 && -10^10 < x < 10^10, {x, y}, Integers] // Timing
```

```
{0.475418 Second, (x == -2 && y == 1) || (x == 0 && y == 1) ||
  (x == 1 && y == 0) || (x == 1 && y == 4) || (x == 2 && y == 1) || (x == 508 && y == 273)}
```

■ Pour obtenir la figure de la lettre

```
Needs["Graphics`ImplicitPlot`"]
Show[ImplicitPlot[1 + 12 x + 2 x^2 + 7 y + 5 x*y + 2 y^2 == 0,
   {x, -5, 4.5}, PlotRange → {-10, 0}, AxesLabel → {"x", "y"}, TextStyle →
    {FontFamily -> "Lucida Grande", FontSlant → "Italic", FontSize → 12},
   DisplayFunction → Identity], Graphics[
   {Hue[1], PointSize[0.025], Map[Point, {{-4, -1}, {2, -3}, {4, -9}}]}],
  DisplayFunction → $DisplayFunction];
```